F07HRF (CPBTRF/ZPBTRF) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07HRF (CPBTRF/ZPBTRF) computes the Cholesky factorization of a complex Hermitian positive-definite band matrix.

2 Specification

```
SUBROUTINE F07HRF(UPLO, N, KD, AB, LDAB, INFO)
ENTRY      cpbtrf(UPLO, N, KD, AB, LDAB, INFO)
INTEGER           N, KD, LDAB, INFO
complex           AB(LDAB,*)
CHARACTER*1       UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the Cholesky factorization of a complex Hermitian positive-definite band matrix A either as $A = U^H U$ if UPLO = 'U', or $A = LL^H$ if UPLO = 'L', where U (or L) is an upper (or lower) triangular band matrix with the same number of super-diagonals (or sub-diagonals) as A.

4 References

[1] Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville

[2] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether the upper or lower triangular part of A is stored and how A is factorized, as follows:

if UPLO = 'U', then the upper triangular part of A is stored and A is factorized as $U^H U$, where U is upper triangular;

if UPLO = 'L', then the lower triangular part of A is stored and A is factorized as LL^H, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: KD — INTEGER *Input*

On entry: k, the number of super-diagonals or sub-diagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB,$*$) — ***complex*** array *Input/Output*

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the n by n Hermitian positive-definite band matrix A, stored in rows 1 to $k+1$. More precisely, if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element $a_{i,j}$ in AB$(k+1+i-j,j)$ for $\max(1,j-k) \leq i \leq j$; if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element $a_{i,j}$ in AB$(1+i-j,j)$ for $j \leq i \leq \min(n,j+k)$.

On exit: the upper or lower triangle of A is overwritten by the Cholesky factor U or L as specified by UPLO, using the same storage format as described above.

5: LDAB — INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HRF (CPBTRF/ZPBTRF) is called.

Constraint: LDAB $\geq$ KD + 1.

6: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i is not positive-definite and the factorization could not be completed. Hence A itself is not positive-definite. This may indicate an error in forming the matrix A. There is no routine specifically designed to factorize a Hermitian band matrix which is not positive-definite; the matrix must be treated either as an nonsymmetric band matrix, by calling F07BRF (CGBTRF/ZGBTRF) or as a full Hermitian matrix, by calling F07MRF (CHETRF/ZHETRF).

7 Accuracy

If UPLO = 'U', the computed factor U is the exact factor of a perturbed matrix $A+E$, where

$$|E| \leq c(k+1)\epsilon|U^H||U|,$$

$c(k+1)$ is a modest linear function of $k+1$, and ϵ is the ***machine precision***. If UPLO = 'L', a similar statement holds for the computed factor L. It follows that $|e_{ij}| \leq c(k+1)\epsilon\sqrt{a_{ii}a_{jj}}$.

8 Further Comments

The total number of real floating-point operations is approximately $4n(k+1)^2$, assuming $n \gg k$.

A call to this routine may be followed by calls to the routines:

F07HSF (CPBTRS/ZPBTRS) to solve $AX = B$;

F07HUF (CPBCON/ZPBCON) to estimate the condition number of A.

The real analogue of this routine is F07HDF (SPBTRF/DPBTRF).

9 Example

To compute the Cholesky factorization of the matrix A, where

$$A = \begin{pmatrix} 9.39+0.00i & 1.08-1.73i & 0.00+0.00i & 0.00+0.00i \\ 1.08+1.73i & 1.69+0.00i & -0.04+0.29i & 0.00+0.00i \\ 0.00+0.00i & -0.04-0.29i & 2.65+0.00i & -0.33+2.24i \\ 0.00+0.00i & 0.00+0.00i & -0.33-2.24i & 2.17+0.00i \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07HRF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KMAX, LDAB
      PARAMETER        (NMAX=8,KMAX=8,LDAB=KMAX+1)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         cpbtrf, X04DFF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07HRF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
*
*        Factorize A
*
         CALL cpbtrf(UPLO,N,KD,AB,LDAB,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
```

```
*           Print factor
*
            IFAIL = 0
            IF (UPLO.EQ.'U') THEN
*
               CALL X04DFF(N,N,0,KD,AB,LDAB,'Bracketed','F7.4','Factor',
     +                     'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
*
            ELSE IF (UPLO.EQ.'L') THEN
*
               CALL X04DFF(N,N,KD,0,AB,LDAB,'Bracketed','F7.4','Factor',
     +                     'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
*
            END IF
*
         ELSE
            WRITE (NOUT,*) 'A is not positive-definite'
         END IF
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07HRF Example Program Data
  4  1                                                      :Values of N and KD
  'L'                                                       :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
```

9.3 Program Results

```
 F07HRF Example Program Results

 Factor
                   1                  2                  3                  4
 1  ( 3.0643, 0.0000)
 2  ( 0.3524, 0.5646) ( 1.1167, 0.0000)
 3                    (-0.0358,-0.2597) ( 1.6066, 0.0000)
 4                                      (-0.2054,-1.3942) ( 0.4289, 0.0000)
```